Constrained minimization

Maximize the volume (x=i, y=R)

```
> f := -subs(SUBS,x*y*z) ;
```
Maximize the volume

$$f := -x\,y\,z \tag{1}$$

Constraints

```
> h  := S-2*(x*y+x*z+y*z) ;
  g1 := x >= 0 ;
  g2 := y >= 0 ;
  g3 := z >= 0 ;
```

$$h := -2\,x\,y - 2\,x\,z - 2\,y\,z + S$$
$$g1 := 0 \leq x$$
$$g2 := 0 \leq y$$
$$g3 := 0 \leq z \tag{2}$$

```
> L := f-lambda*h -mu1*rhs(g1) -mu2*rhs(g2) -mu3*rhs(g3) ;
```

$$L := -x\,y\,z - \lambda\,(-2\,x\,y - 2\,x\,z - 2\,y\,z + S) - \mu 1\,x - \mu 2\,y - \mu 3\,z \tag{3}$$

Nonlinear system

```
> EQ1 := diff(L,x) ;
  EQ2 := diff(L,y) ;
  EQ3 := diff(L,z) ;
  EQ4 := h ;
  EQ5 := rhs(g1)*mu1 ;
  EQ6 := rhs(g2)*mu2 ;
  EQ7 := rhs(g3)*mu3 ;
```

$$EQ1 := -y\,z - \lambda\,(-2\,y - 2\,z) - \mu 1$$
$$EQ2 := -x\,z - \lambda\,(-2\,x - 2\,z) - \mu 2$$
$$EQ3 := -x\,y - \lambda\,(-2\,x - 2\,y) - \mu 3$$
$$EQ4 := -2\,x\,y - 2\,x\,z - 2\,y\,z + S$$
$$EQ5 := \mu 1\,x$$
$$EQ6 := \mu 2\,y$$
$$EQ7 := \mu 3\,z \tag{4}$$

```
> SOL := solve( {EQ||(1..7),x>0,y>0,z>0,mu1>=0,mu2>=0,mu3>=0}, {x,y,
  z,lambda,mu1,mu2,mu3} );
```

$$SOL := \left\{\lambda = -\frac{1}{24}\sqrt{6}\sqrt{S}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = -\frac{1}{6}\sqrt{6}\sqrt{S}, y = -\frac{1}{6}\sqrt{6}\sqrt{S}, z = -\frac{1}{6}\sqrt{6}\sqrt{S}\right\}, \left\{\lambda = \frac{1}{24}\sqrt{6}\sqrt{S}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = \frac{1}{6}\sqrt{6}\sqrt{S}, y = \frac{1}{6}\sqrt{6}\sqrt{S}, z = \frac{1}{6}\sqrt{6}\sqrt{S}\right\} \tag{5}$$

```
> SOL := SOL[2] ;
```

$$SOL := \left\{\lambda = \frac{1}{24}\sqrt{6}\sqrt{S}, \mu 1 = 0, \mu 2 = 0, \mu 3 = 0, x = \frac{1}{6}\sqrt{6}\sqrt{S}, y = \frac{1}{6}\sqrt{6}\sqrt{S}, z \tag{6}$$

$$= \frac{1}{6}\sqrt{6}\sqrt{S}\}$$

Check if the solution is a mimimum. Compute the gradient of the active constraints

```
> subs(SOL,rhs(g1)) ; # g1 is not active
  subs(SOL,rhs(g2)) ; # g2 is not active
  subs(SOL,rhs(g3)) ; # g3 is not active
```

$$\frac{1}{6}\sqrt{6}\sqrt{S}$$

$$\frac{1}{6}\sqrt{6}\sqrt{S}$$

$$\frac{1}{6}\sqrt{6}\sqrt{S} \tag{7}$$

```
> G := subs(SOL,<diff(h,x)|diff(h,y)|diff(h,z)>) ;
```

$$G := \begin{bmatrix} -\frac{2}{3}\sqrt{6}\sqrt{S} & -\frac{2}{3}\sqrt{6}\sqrt{S} & -\frac{2}{3}\sqrt{6}\sqrt{S} \end{bmatrix} \tag{8}$$

```
> K := op(LinearAlgebra[NullSpace](G)) ;
  K := <K[1]|K[2]>;
```

$$K := \begin{bmatrix} -1 \\ 0 \\ 1 \end{bmatrix}, \begin{bmatrix} -1 \\ 1 \\ 0 \end{bmatrix}$$

$$K := \begin{bmatrix} -1 & -1 \\ 0 & 1 \\ 1 & 0 \end{bmatrix} \tag{9}$$

The hessian of the Lagrangian

```
> H := <<diff(L,x,x),diff(L,x,y),diff(L,x,z)>|
        <diff(L,x,y),diff(L,y,y),diff(L,y,z)>|
        <diff(L,x,z),diff(L,z,y),diff(L,z,z)>>;
  H := subs(SOL,H) ;
```

$$H := \begin{bmatrix} 0 & -z+2\lambda & -y+2\lambda \\ -z+2\lambda & 0 & -x+2\lambda \\ -y+2\lambda & -x+2\lambda & 0 \end{bmatrix}$$

$$H := \begin{bmatrix} 0 & -\frac{1}{12}\sqrt{6}\sqrt{S} & -\frac{1}{12}\sqrt{6}\sqrt{S} \\ -\frac{1}{12}\sqrt{6}\sqrt{S} & 0 & -\frac{1}{12}\sqrt{6}\sqrt{S} \\ -\frac{1}{12}\sqrt{6}\sqrt{S} & -\frac{1}{12}\sqrt{6}\sqrt{S} & 0 \end{bmatrix} \tag{10}$$

```
> HReduced := LinearAlgebra[Transpose](K).H.K ;
```

(11)

$$HReduced := \begin{bmatrix} \frac{1}{6}\sqrt{6}\sqrt{S} & \frac{1}{12}\sqrt{6}\sqrt{S} \\ \frac{1}{12}\sqrt{6}\sqrt{S} & \frac{1}{6}\sqrt{6}\sqrt{S} \end{bmatrix} \quad (11)$$

Use Sylvester criteria

```
> LinearAlgebra[Determinant](HReduced[1..1,1..1]) ;
```

$$\frac{1}{6}\sqrt{6}\sqrt{S} \quad (12)$$

```
> LinearAlgebra[Determinant](HReduced[1..2,1..2]) ;
```

$$\frac{1}{8}S \quad (13)$$